

Chapter 2

Simple Economic Models

Abstract This chapter illustrates two models using *Maxima*. The first model is the production possibilities curve, which is a simple model of economy-wide production. The second model is that of competitive markets, where demand and supply determine the equilibrium price and quantity of a good. We also extend this model to examine disequilibrium and the effects of shifts in demand or supply, along with the relevance of elasticities, the impact of taxes, and the value that markets generate.

2.1 Production Possibilities

Perhaps the most abstract model of an economy is the production possibilities curve (PPC), which reduces the economy to two goods and places only quite general conditions on the functional relationship between the two goods. As abstract as it is, this model offers important lessons.

2.1.1 An Illustrative Example

We impose a specific functional form on the PPC, $y = a + b \cdot x^{2.5}$, where $a > 0$ and $b < 0$. We state this as an *explicit function*, so that y appears as a dependent variable. This representation is for convenience only. The levels of x and y are jointly determined, and their values must satisfy this equation: $y - (a + b \cdot x^{2.5}) = 0$.

The model appears below. Before graphing the function, we determine the range over which nonnegative values x correspond to nonnegative values of y , because only points in the first quadrant make economic sense. For this purpose, we use `find_root`. We enter the function and the range over which *Maxima* is to search. We use a quite high maximum value.¹

¹If this value is set too low, an error message will occur, showing the values of y at the endpoints. Executing the command again with an increased maximum value of x will lead to a solution.

The output generated by this set of commands shows the original function and the value of x at which y is closest to zero.


```
[f(x,a,b):=a+b*x^2.5,[a1,b1]:[500,-0.01],
x1max:find_root(f(x,a1,b1),0,1000)];
```

```
[f(x,a,b):=a+b*x^2.5,[500,-0.01],75.786]
```

The `wxdraw2d` command below plots a single explicit function, the PPC. The x and y ranges are set at 1.1 times the maximum value of x (`x1max` from above) and of y (`a1`). Inside the `explicit` command, x ranges from 0 to `x1max`.

Examining the graph points out the first lesson from this simple model: some levels of output cannot be attained given the resource supplies and technology that define this PPC. Even if all resources are used (no unemployment) and they are used efficiently, all points above the curve are ruled out.

```
wxdraw2d
(xlabel="Good x",
ylabel="Good y",
yrange=[0,1.1*a1],
xrange=[0,
1.1*x1max],
line_width=2,
explicit(f(
x,a1,b1),
x,0,x1max))$
```



2.1.2 Opportunity Costs

The second implication of the graph is the existence of trade-offs. Once the PPC has been reached, more x can be had only at the cost of less y and vice versa. Opportunity costs exist.

The next cell evaluates the function $f(x, a1, b1)$ for four values of x . The first input line defines the initial values $x1$ and $x2$, and the change in x from each of these starting points Δx .² Then values for y are returned. Thus, the function is evaluated at approximately these values of x : 15.16, 18.95, 37.89, and 41.68. The resulting values of y are approximately 491.1, 484.4, 411.6, and 387.8. As x increases, y decreases, reflecting the opportunity cost of increased x production.


²The values are defined in terms of `x1max`. This is not necessary. Specific numbers can be entered if you prefer, as long as the values are within the range determined above. The changes are discrete, so the Greek letter Delta, Δ is used.

```
[[x1, x2, Deltax] : [x1max/5, x1max/2, x1max/20],
[y1, y1d] : [f(x1, a1, b1), f(x1 + Deltax, a1, b1)],
[y2, y2d] : [f(x2, a1, b1), f(x2 + Deltax, a1, b1)]];

[[15.157, 37.893, 3.7893], [491.06, 484.38], [411.61, 387.83]]
```

The next graph adds these four points to the PPC. The graph shows that the decrease in y production increases as we move from $x = x_1$ to $x = x_2$. That is, this PPC displays the *increasing* opportunity cost of x as x increases.³

```
wxdraw2d(xlabel="Good x",ylabel="Good y",yrange=
[0,1.1*a1],xrange=[0,1.1*x1max],line_width=2,
explicit(f(x,a1,b1),x,0,x1max),point_size=2,
point_type=asterisk,points([x1,x1+Deltax,x2,
x2+Deltax],[y1,y1d,y2,y2d]))$
```



The cell below calculates the opportunity cost of x over the ranges considered above. Note the negative sign: Decreased y is a positive cost.

```
oc1: -(y1d-y1)/(Deltax)$ oc2: -(y2d-y2)/(Deltax)$
["OpportunityCosts, 2 Ranges:", oc1, oc2];
```

```
[Opportunity Costs, 2 Ranges:, 1.7631, 6.276]
```

The graph below shows opportunity cost as a continuous function of x . The opportunity cost at any point on the PPC is the negative of the PPC's slope at that point. The `diff` command takes the first derivative of the function, providing a functional specification of the slope.

Developing this function requires two steps, both of which appear on the first input line below. First apply `diff`. Then define the function as the result of the `diff` operation. The "quote - quote" before the `%` tells *Maxima* to evaluate the result of taking the derivative, and not just to keep it in memory as an expression. Again, note the negative sign.


³Economic theory does not dictate that opportunity cost must increase. The simple Ricardian model of comparative advantage, with a linear PPC, exhibits constant opportunity cost. PPCs can bow in, at least over some range, so that opportunity cost decreases.

The graph shows two points, each at the midpoint of the ranges defined above.⁴ We extend the graph just beyond the largest value defined above, to $x_2 + 2 \cdot \Delta x$.

The graph shows that the PPC above generates continuously increasing opportunity cost. It also shows that at the two points on the graph, the result of computing opportunity cost over a discrete range and the result of computing it at a point are quite close to each other, for this function.⁵

```
g(x, a1, b1) := -''(diff(f(x, a1, b1), x));
wxdraw2d(ylabel="Good y per unit of Good x",title =
  "Opportunity Costs: Function and Points",xlabel=
  "Good x",explicit(g(x, a1,b1),x,0,x2+2*Deltax),
  point_size=2,points([(x1+Deltax/2),(x2+Deltax/2)],
    [oc1,oc2]))$
```

$$g(x, a1, b1) := -(-0.025 x^{1.5})$$



2.1.3 Shifting PPC

A change in either the resource availability or technology shifts the PPC. An increased resource base or an improved technology would shift the PPC outward (more y for each level of x produced). Whether the shift would be parallel or, more likely, take some other form depends on the details.

The illustration below shows a special case. Either technology is limited to x production or only resources used in x production increases. When all resources are allocated to y production, the quantity does not change.


Before drawing the graph, use `find_root` to set the range of x (to limit the output to nonnegative y values). The output includes both the initial x limit and its

⁴ $(x_1 + x_1 + \Delta x)/2 = x_1 + \Delta x/2$

⁵Evaluating the oc function at $x = x_1 + \Delta x/2$ and at $x = x_2 + \Delta x/2$ yields values that differ from the values computed over the ranges only at the third decimal point. These values are 1.76 and 6.274, respectively.

new counterpart. For graphing, we define y_{\max} as the maximum of two values, a_1 and a_2 . In the case below, a does not change so both values are equal. We define x_{\max} as the larger of $x_{1\max}$ and $x_{2\max}$.

```
[a2,b2]:[500,-0.0075]$ x2max:find_root(f(x,a2,b2),
0,1000)$ [x1max,x2max]; ymax:max(a1,a2)$
xmax:max(x1max,x2max)$
wxdraw2d(title="PPC shift: lower b value",xlabel=
"Good x",ylabel="Good y", yrange=[0,1.1*ymax],
xrange=[0,1.1*xmax],key="Initial PPC",line_width=2,
explicit(f(x,a1,b1),x,0,xmax),color=gray,key=
"New PPC",explicit(f(x,a2,b2),x,0,xmax))$
```



In the example depicted, the opportunity cost of x (the number of units of y given up per increased unit of x) falls, shifting the PPC out to the gray line, which has a smaller slope at each x value than the initial black line.⁶

Chapter 12 revisits the PPC, deriving a PPC from microeconomic foundations. In doing so, it investigates the nature of efficiency, identifying the conditions that are required for an economy to reach its PPC. It also shows how the price system can guide an economy toward the PPC.

2.2 Competitive Markets

The cell below contains two representations of the demand curve for a product and two corresponding representations of the supply curve. In each case the quantity equation occurs first, and the *inverse*, or price equation occurs next.

⁶This implies, however, that the opportunity cost of y increases: At each level of x , increasing y by an additional unit requires that more units of x be forgone than before the shift.

2.2.1 Demand Curves and Supply Curves

The demand curve $DQ = a \cdot p^E$ states the quantity demanded at each of a range of prices. The inverse demand curve $DP = (q/a)^{1/E}$ states the height of the supply curve, or the price that buyers are willing and able to pay, at each of a range of quantities. The supply curve can be represented in either of the two ways listed below, as well.

We use DQ to indicate “the demand curve, quantity” and DP to indicate “the demand curve, price.” Likewise for the supply curve. Henceforth, variable names will often consist of a series of letters that indicate the variable’s economic meaning. When two variables are to be multiplied, this text uses the \cdot sign, a counterpart to the $*$ in the *wxMaxima* commands.

The output below restates the input. Often, we suppress such output by ending commands with dollar signs rather than semicolons. We specify the parameter values below, but the following must be true: E must be negative (demand curves slope downward), a must be positive (only positive prices and quantities are allowed), b can be either positive or negative, and c is nonnegative. This last constraint is not mandated by economic theory. Also, if $b < 0$, c must be positive, so that some part of the supply curve lies in the first quadrant.

The first pair of commands creates the demand curve which defines the quantity demanded at each price, and its equivalent inverse demand curve which defines the price that buyers are willing and able to pay for each quantity. The second pair does for the supply curve and its inverse what the second set does for the demand curve and its inverse.⁷

$[DQ(p, a, E) := a \cdot p^E,$	$DP(q, a, E) := (q/a)^{(1/E)}];$
$[SQ(p, b, c) := b + c \cdot p,$	$SP(q, b, c) := (q - b)/c];$

$$[DQ(p, a, E) := a p^E, DP(q, a, E) := (\frac{q}{a})^{\frac{1}{E}}]$$

$$[SQ(p, b, c) := b + c p, SP(q, b, c) := \frac{q-b}{c}]$$

The first command in the next cell evaluates the inverse demand curve given parameter values. It instructs *Maxima* to produce the equation for a specific demand curve. The second command maintains these values of a and E and adds the value of q . It instructs *Maxima* to find a point on the demand curve ($q = 40$, $p \approx 3.8981$). Similar exercises would produce counterparts for supply.

$[DP(q, 1200, -2.5), DP(40, 1200, -2.5)];$	$[\frac{17.048}{q^{0.4}}, 3.8981]$
--------------------------------------------	------------------------------------

⁷We could use *Maxima* to determine the equations for the inverse demand and supply curves. The accompanying exercise set takes this approach.

2.2.2 Equilibrium

We begin by solving the system of equations for the quantity and requiring *Maxima* to provide numerical value(s). The solution involves an expression that combines a polynomial (supply) and a transcendental function (demand).

For a relatively complex equation like this one, *Maxima*'s `solve` function might not suffice, so, we use `find_root`.⁸ Iteration begins at $q = 0.0001$, avoiding division by zero, and ends at 500.⁹ The resulting value is assigned the name `qE0`.

The equilibrium quantity is $qE0 = 35.765$. We determine the resulting price level twice, once by putting $qE0$ into the inverse demand curve and then by putting $qE0$ into the inverse supply curve. The comforting result is that both numbers are equal; the demand price equals the supply price.

```
[a0,E0,b0,c0] : [1200,-2.5,-5,10]$      [qE0:
  find_root(DP(q,a0,E0)-SP(q,b0,c0),q,.001,500),
  ["Prices",DP(qE0,1200,-2.5), SP(qE0,-5,10)]];
```

```
[35.765, [Prices, 4.0765, 4.0765]]
```

Solving the system for quantity is arbitrary. We could have determined the price at which the demand price equals the supply price.

```
[pE0:find_root(DQ(q,a0,E0)-SQ(q,b0,c0),q,0.001,10),
  ["Quantities", DQ(pE0,1200,-2.5), SQ(pE0,-5,10)]  ];
```

```
[4.0765, [Quantities, 35.765, 35.765]]
```

Next, we create a list of eight prices, each stated as a multiple of the equilibrium price, and the associated quantities demanded, quantities supplied, and excess demand values. To do this we create two new equations, an equation for a specific demand curve and its counterpart for supply.

Maxima's `makelist` command is used to create a list of prices, `pList`. The prices range from $\frac{pE0}{5}$ to $\frac{8 \cdot pE0}{5}$. With this list in place, we use `maplist` to map the demand curve and supply curve equations onto the list. The results are `qdList0` and `qsList0`. For each, we map the relevant function onto the list. Once the lists are created, we use the `append` command to add names. The excess demand list is created by subtraction.

⁸Actually, `solve` does work in this case, but it yields four solutions, three of which involve imaginary numbers or negative values. The fourth is the positive real solution that `find_root` provides directly.

⁹If the upper bound had been set too low, `find_root` would have returned an error message because the difference between the demand price and the supply price would have been positive at both ends of the range. If this happens, increase the upper end of the range and repeat the command. Or draw a simple graph to determine the approximate value.

```
[DQ0(p) := '' (DQ(p,a0,E0)), SQ0(p) := '' (SQ(p,b0,c0))];
/*Create lists*/ pList0:makelist(pE0*i/5,i,1,8)$
qdList0:maplist(DQ0,pList0) $
qsList0 : maplist(SQ0, pList0) $
/*Append names */ pList0:append(["Price"],pList0)$
qdList0:append( ["DQ"], qdList0)$ qsList0:
append(["SQ"],qsList0)$ edList0:qdList0-qsList0$
/*Create table */
transpose(matrix(pList0,qdList0,qsList0,edList0));
```

$$[DQ0(p) := \frac{1200}{p^{2.5}}, \quad SQ0(p) := 10p - 5]$$

Price	DQ	SQ	DQ-SQ
0.815	1999.3	3.153	1996.2
1.6306	353.44	11.306	342.13
2.4459	128.26	19.459	108.8
3.2612	62.479	27.612	34.867
4.0765	35.765	35.765	$-2.13163 \cdot 10^{-14}$
4.8918	22.673	43.918	-21.245
5.7071	15.422	52.071	-36.649
6.5224	11.045	60.224	-49.179


The table shows that for low prices, a quite large excess demand results. This reflects the fact that the demand curve selected is quite price elastic over its range. At $Price = 4.0765$, the excess demand is zero.¹⁰

Finally we draw the supply and demand curves and add the line that is defined above. Each of the two relationships involves an `explicit` command within `wxdraw2d`. The y range is limited to positive values and is truncated at twice the equilibrium price in order to emphasize the relevant part of the demand curve. Inside the `explicit` commands for the demand and supply curves, we limit the range to $1.5 \cdot qE0$.

To draw a line that highlights the equilibrium price and quantity, we use an `if ... then ... else` command. This draws a horizontal line for $q < qE0$ and has the line drop to zero at $q = qE0$. This function is drawn from $q = 0$ to $q = 1.05 \cdot qE0$. Notice that each individual curve must be drawn over its own specified range and that these ranges can differ among curves.

¹⁰Rounding error resulting from the use of a numerical method to find the equilibrium price results in the reported value, which is nearly zero.


```
eqline:if q < qE0 then pE0 else 0$ wxdraw2d( yrange=
[0, 2.0*pE0], line_type=dots,ylabel="$ per unit",
xlabel="Quantity: units per time period",explicit(
eqline,q,0,1.05*qE0),line_width=2,line_type=solid,
key="Demand",explicit(DP(q,a0,E0),q,0,1.5*qE0),
key="Supply",color=gray,
explicit(SP(q,b0,c0),q,0,1.5*qE0))$
```



2.2.3 Disequilibrium

Confirm that for $q > qE0$ the supply curve lies above the demand curve. That is, confirm that the price that producers would have to be paid in order to produce this quantity exceeds the price that buyers are willing and able to pay for that quantity.

```
[1.1*qE0, [DP(1.1*qE0,1200,-3),SP(1.1*qE0,20,5)]];
[39.342, [3.1245, 3.8683]]
```

Confirm that for $q < qE0$, the demand curve lies above the supply curve.

```
[0.9*qE0, [DP(0.9*qE0,1200,-3),SP(0.9*qE0,20,5)]];
[32.189, [3.3406, 2.4377]]
```

2.2.4 Price Ceilings

Markets tend to drive the price toward the equilibrium price. Market tendencies can be overcome, however, by government policies that mandate a price other than the one that clears the market.


The cell below specifies that the price may not rise above $0.8 \cdot pE0$. (The value of multiple can be changed.) The second input line determines the quantity

demanded and the quantity supplied at the mandated price and assigns names to these values. For graphing, the maximum of these two values is assigned the name `maxq`.

```
multiple:0.8$ [pNE:multiple*pE0,
[qNED,qNES]:
[DQ(pNE,a0,E0),SQ(pNE,b0,c0)]];
maxq : max(qNED, qNES)$ [3.2612, [62.479, 27.612]]
```

The graph below shows the demand curve, the supply curve, and the two points derived above. The horizontal distance between the two quantities is the per-period shortage. The x and y ranges are set as multiples of the equilibrium p and x values, respectively, to allow for changing the value of `multiple`.¹¹

```
wxdraw2d(yrange=[0,2.5*pE0],xrange=[0,1.25*maxq],
xlabel="Quantity in units per time period",ylabel=
"$ per unit",line_width=2,key="Demand",explicit(
DP(q,a0,E0),q,.01,1.25*maxq),key="Supply",color=
gray,explicit(SP(q,b0,c0),q,0,1.25*maxq),
point_size=2,points_joined=true,color=black,
line_type=dots,key="Excess Demand",
points([qNED, qNES],[pNE,pNE]))$
```



Sometimes black markets result from price ceilings. How a black market works (which price and quantity pair tends to result) depends on the specifics of the case. Suppose that black market sellers can buy the 27.612 units and auction them to the

¹¹If `multiple > 1`, then a price floor is established. An exercise using these demand and supply curves but analyzing a price floor appears in the online exercises.

highest bidders. Then the solution is simple: the black market price is \$4.521 per unit. We determine this by placing the quantity q_{NES} into the demand curve.¹²

$$p_{BM} : DP(q_{NES}, a_0, E_0); \quad 4.521$$

2.2.5 Shifts

Supply and Demand shifts are easily modeled numerically and graphically using the techniques developed in the previous section. The first input group shows the coefficients of the initial demand and supply curves, used above, and the coefficients that result in shifted demand and supply curves.

$$\begin{array}{l} [a_0, E_0, b_0, c_0] : [1200, -2.5, -5, 10] \$ \\ [a_0, E_1, b_1, c_1] : [1200, -1.5, -5, 15] \$ \end{array}$$

2.2.5.1 Demand Shift

Shift the demand curve. Either a or E could be changed. We leave a unchanged and change E from $E = -2.5$ to $E = -1.5$. That is, the demand curve becomes less responsive to price (elasticity falls).

The resulting equilibrium values appear below: $qE1 = 83.597$ and $pE1 = 5.9065$. Both quantity and price increase relative to their original values, $qE0 = 48.939$ and $pE0 = 3.5959$, which is to be expected as the demand curve shifts to the right.

It might not be apparent that decreasing price responsiveness (making the demand curve less elastic) should shift the curve to the right. In fact, it does so only for prices above $p = 1$. At $p = 1$, the quantity equals a , whatever the elasticity value. Thus, the two curves intersect at $p = 1$. (Exercise: Confirm this assertion graphically.) The relevant price range is higher for the new demand curve, and the quantity demanded falls less along the new demand curve than along the initial one.


$$\begin{array}{l} qE1 : \text{find_root}(DP(q, a_0, E_1) - SP(q, b_0, c_0), q, .001, 500) \$ \\ pE1 : DP(qE1, a_0, E_1) \$ \\ \text{matrix}(["qE0", "qE1", "pE0", "pE1"], [qE0, qE1, pE0, pE1]); \end{array}$$

$$\begin{bmatrix} qE0 & qE1 & pE0 & pE1 \\ 35.765 & 64.914 & 4.0765 & 6.9914 \end{bmatrix}$$

¹²A more likely outcome is that some units will sell in the black market for a price higher than this mandated price, some will sell for a price between the free-market equilibrium price and the mandated price, and still others will sell at the mandated price.

The graph shows the shift in the demand curve and its effects. The new demand curve lies above (to the right of) the initial one, so both price and quantity tend to increase. The graph is the same as in the previous section with a few minor changes. First, variables $maxq$ and $maxp$ are defined for graphing purposes. One more explicit command is added.

```
maxq : max(qE0, qE1)$ maxp : max(pE0, pE1)$
wxdraw2d( user_preamble="set key bottom center",
  yrange=[0,1.5*maxp],ylabel="$ per unit",
  xlabel="Quantity: units per time period",key=
  "Initial Demand",explicit(DP(q,a0,E0),q,0,
  1.5*maxq),line_width=2,key="New Demand",
  explicit(DP(q,a0,E1),q,0,1.5*maxq),color=gray,
  key="Supply",explicit(SP(q,b0,c0),q,0,1.5*maxq))$
```



2.2.5.2 Supply Shift

The input cell below shows the effect of an increased supply. The table compares the initial quantity and price with the quantity and price after the shift in supply.

```
qE2:find_root(DP(q,a0,E0)-SP(q,b0,c1),q,.001,500)$
pE2 : DP(qE2,a0,E0)$
matrix(["qE0","qE2","pE0","pE2"],[qE0,qE2,pE0,pE2]);
```

qE0	qE2	pE0	pE2
35.765	48.939	4.0765	3.5959

As an exercise, replicate the graph above, with a supply increase replacing a demand increase. Also, as further exercises, graphically analyze the effects of simultaneous demand and supply shifts. Confirm that when both curves shift, the direction of change is determinate for either price or quantity, and that the direction depends on the specific shift sizes for the other variable.

2.3 Elasticities

2.3.1 Price Elasticity

The responsiveness of the quantity demanded to price changes can be measured in terms of either slope or elasticity. The *slope* of the demand curve ($\frac{dq}{dp}$) is the number of units that quantity changes per one-unit change in price. The *price elasticity* is the percentage change in quantity per 1 % change in price.

Each measure has its merits. For a given product in a given setting, the slope is quite informative. For more general statements, elasticity becomes more useful, for two main reasons. First, it is free of units, so elasticities of demand for various products can be compared. Second, the elasticity is closely related to how total expenditures for a good change in response to a price change. This is particularly important for analyzing the behavior of price-searching firms (e.g., monopoly, monopolistic competition).

2.3.1.1 Constant-Elasticity Demand Curve

This section begins with the demand curve that has been used above. This demand curve exhibits constant elasticity. That is, the price elasticity of demand is the same at all points on the demand curve.

The demand curve appears twice in the input group below. It is evaluated for $a = 1200$ and $E = -2.5$ at two prices, $p = 9.9$ and $p = 10.1$. We use the midpoint formula $E_{arc} = \frac{q_2 - q_1}{q_{mean}} / \frac{p_2 - p_1}{p_{mean}}$.


The value of p_{mean} is 10 and the change in p is 0.2, so these values are entered directly. Their q counterparts are computed. The resulting “arc” elasticity is quite close to the “point” elasticity, -2.5 . As an exercise, enter another pair of price values and confirm that the “arc” elasticity is nearly -2.5 .

[[q1, q2] : [DQ(9.9, 1200, -2.5), DQ(10.1, 1200, -2.5)],
 ((q2 - q1) / (0.5 * (q2 + q1))) / (0.2 / 10)] ;

[[3.8913, 3.7015], -2.4996]

The graph below shows three constant-elasticity demand curves with different price elasticities. The first is elastic ($E = -2.5$), the second is unit elastic ($E = -1$), and the third is inelastic ($E = -0.25$). The three curves intersect when $p = 1$, with $q = a = 1,200$, as noted earlier.

```
wxdraw2d(yrange=[0,3],xlabel="Quantity",ylabel=
"$ per unit",user_preamble="set key bottom left",
key="Elastic",explicit(DP(q,1200,-2.5),q,1,1500),
line_width=2,key="Unit Elasticity",explicit(
DP(q,1200,-1),q,1,1500),color=gray,line_width=2,key
="Inelastic",explicit(DP(q,1200,-0.25),q,1,1500))$
```



2.3.1.2 Elasticity, Price, and Expenditures

The price elasticity of demand determines the relationship between price changes and corresponding changes in total spending on the product or, equivalently, total revenue for the seller(s) of the product. When the demand curve is price elastic, price and total spending are inversely related; when the demand curve is price inelastic, price and total spending change in the same direction; and when the demand curve's elasticity is -1.0 , total spending is the same at each price.

The input group below determines the marginal revenue (the change in total spending, or revenue, per one-unit change in quantity). The first statement informs *Maxima* that the price for which the product sells depends on the quantity. Then total revenue is defined.

Finally, marginal revenue is defined as the first derivative of total revenue.¹³

The result shows that $MR = p + \frac{dp}{dq} \cdot q$. Factoring p from this expression yields $MR = p \cdot (1 + (\frac{dp}{dq} \cdot \frac{q}{p}))$, which is equivalent to $MR = p \cdot (1 + \frac{1}{E_p})$, where E_p is the price elasticity. Recall that $E_p < 0$, so demand is elastic when $E_p < -1$ and inelastic when $-1 < E_p \leq 0$.¹⁴

¹³We use binding rather than functional expressions because, in this setting, doing so is simpler. Binding the result of the `diff` command is somewhat easier than expressing the relationship as a function.

¹⁴Be aware that many textbooks, by tradition, use the absolute value of own-price elasticity of demand, so that it is reported as a positive number.

Given that $MR = p \cdot (1 + \frac{1}{E_p})$, when the demand curve is price-elastic, $1 + \frac{1}{E_p}$ must be positive, so quantity and total spending move in the same direction, implying that price and total spending are inversely related. Likewise, price-inelastic implies that price and total spending move in the same direction. Finally, if $E_p = -1$, $MR = 0$, so quantity and total spending are unrelated, as are price and total spending.

$$[\text{depends}(p, q), \quad TR: p \star q, \quad MR: \text{diff}(TR, q)] ;$$

$$[[p(q)], \quad p q, \quad \left(\frac{d}{dq} p\right) q + p]$$

The input/output group below shows that for the constant-elasticity curve defined above, the relationship defined above holds. To make the result of the `diff` command easily intelligible requires use of the `subst` and `factor` commands. The output shows that $MR = p + \frac{p}{E}$ —so $MR = p \cdot (1 + \frac{1}{E})$ —or $MR = p \cdot \frac{E+1}{E}$.

$$[TR(q, a, E) := q \star (q/a)^{(1/E)}, \quad MR: \text{diff}(TR(q, a, E), q), \\ \text{subst}(p, (q/a)^{(1/E)}, \text{factor}(MR))] ;$$

$$[TR(q, a, E) := q \left(\frac{q}{a}\right)^{\frac{1}{E}}, \quad \frac{\left(\frac{q}{a}\right)^{\frac{1}{E}}}{E} + \left(\frac{q}{a}\right)^{\frac{1}{E}}, \quad \frac{p(E+1)}{E}]$$

The figure below shows two demand curves, one elastic and the other inelastic, and the associated marginal revenue curves. To facilitate graphing, we write the marginal revenue curve using the same functional notation that has been used for the demand curves. For all constant-elasticity demand curves, marginal revenue is a constant times price. For elastic curves, this constant is between 0 and 1. For inelastic curves, the constant is negative. In the examples below, $MR = 0.4 \cdot p$ for the elastic demand curve, and $MR = -\frac{1}{3} \cdot p$ for the inelastic demand curve.

$$MR(q, a, E) := ' '(DP(q, a, E) \star (1 + 1/E)) ; \quad \text{wxdraw2d}(\text{yrange} =$$

$$[-2, 5], \text{xaxis} = \text{true}, \text{yrange} = [-5, 15], \text{xlabel} = \text{"Quantity"},$$

$$\text{ylabel} = \text{"\$ per unit"}, \text{key} = \text{"Demand, Elastic"}, \text{explicit}(\text{DP}(q, 1200, -2.5), q, 1, 1500), \text{key} = \text{"MR, Elastic Demand"},$$

$$\text{color} = \text{gray}, \text{explicit}(MR(q, 1200, -2.5), q, 1, 1500), \text{color} =$$


$$\text{black}, \text{line_width} = 2, \text{key} = \text{"Demand, Inelastic"},$$

$$\text{explicit}(DP(q, 1200, -0.5), q, 1, 1500), \text{color} = \text{gray},$$

$$\text{key} = \text{"MR, Inelastic Demand"},$$

$$\text{explicit}(MR(q, 1200, -0.75), q, 1, 1500)) \$$$

$$MR(q, a, E) := \left(\frac{q}{a}\right)^{\frac{1}{E}} \left(\frac{1}{E} + 1\right)$$



2.3.1.3 Linear Demand Curve

Linear demand curves have constant slopes and varying elasticities. The analysis below shows that $E_p = \frac{p}{b \cdot q}$. Thus, the elasticity is a negative value that approaches zero (the demand curve becomes less elastic) as q increases.

$$\begin{aligned} &[p : a + b \cdot q, TR : \text{expand}(p \cdot q), \\ &MR : \text{diff}(TR, q), E : (p/q) / (\text{diff}(p, q))]; \end{aligned}$$

$$[bq + a, bq^2 + aq, 2bq + a, \frac{bq+a}{bq}]$$

The demand curve has a point of unitary elasticity when $q = 0.5 \cdot q_{max}$, where q_{max} is the quantity demanded when $p = 0$. Thus, the linear demand curve is elastic for quantities less than one-half of the maximum quantity and inelastic for quantities greater than this value.

$$\text{subst}(q = 0.5 \cdot (-a/b), E); \quad -1.0$$

The example below illustrates these conclusions. This linear demand curve is named pL. Total revenue, marginal revenue, and elasticity are named TRL, MRL, and EL.


$$\begin{aligned} &[pL : 400 - 2.5 \cdot q, \quad TRL : \text{expand}(pL \cdot q), \\ &MRL : \text{diff}(TRL, q), EL : -0.4 \cdot (pL/q)]; \\ &[400 - 2.5q, 400q - 2.5q^2, 400 - 5.0q, -\frac{0.4(400 - 2.5q)}{q}] \end{aligned}$$

We graph the demand curve and the marginal revenue curves in one panel and the elasticity function separately. Elasticity is a pure, unit-free number, so it can be used to compare price elasticities across products. The graphs confirm that the elasticity equals -1 at the midpoint on the demand curve (where $q = 80$). Also, at that point $MRL = 0$.


```

D_MR:gr2d(xlabel="Quantity", ylabel = "$ per unit",
  xaxis=true,color=black,key="Demand",explicit(
    pL,q,0,160),line_width=2, key = "MR", explicit(
    MRL,q,0,160),ytics={-200,0,200},xtics={0,80,160})$
Elasticity:gr2d(user_preamble="set key bottom right",
  xlabel="Quantity",ylabel="No units",yrange=[-5,0],
  ytics={["-5.0",-5],["-1.0",-1],["0.0",0]},
  color=black,explicit(-1,q,0,160),color=gray40,key=
  "Elasticity",explicit(EL,q,0,160),xtics={0,80,160})$
wxdraw(D_MR, Elasticity,dimensions=[360,360])$

```



2.3.2 Other Elasticities

The own-price elasticity of demand is one of a number of elasticities that can provide useful information. This section considers some others:

- **Income Elasticity of Demand:** The percentage change in the quantity demanded for this product per 1 % change in income.
- **Cross-Price Elasticity of Demand:** The percentage change in the quantity demanded for this product per 1 % change in the price of some other related product.
- **Price Elasticity of Supply:** The percentage change in the quantity supplied per 1 % change in price.

In each case, “other things equal” assumption should be kept in mind.¹⁵

¹⁵Also keep in mind that elasticity is a quite general concept. Although economists define some other elasticities, such as output elasticity and cost elasticity, one should think of an elasticity as

We illustrate the elasticity concepts with constant-elasticity demand and supply functions. First, the demand curve (and its inverse) that appear below are direct extensions of the ones used earlier. Two arguments have been added: the price of a related good (pr) and income (y). Each of the exponents is an elasticity.

$$\begin{aligned} DQ(p, a, ED, pr, EDR, y, EDY) &:= a * p^{ED} * pr^{EDR} * y^{EDY}; \\ DP(q, a, ED, pr, EDR, y, EDY) &:= (q / (a * pr^{EDR} * y^{EDY}))^{1/ED}; \end{aligned}$$

$$\begin{aligned} DQ(p, a, ED, pr, EDR, y, EDY) &:= a p^{ED} pr^{EDR} y^{EDY} \\ DP(q, a, ED, pr, EDR, y, EDY) &:= \left(\frac{q}{a pr^{EDR} y^{EDY}} \right)^{\frac{1}{ED}} \end{aligned}$$

The cell below specifies a set of parameter values. Also, it shows two sets of “other things,” the price of a related good and income, that can shift the demand curve. Then a list of four expressions for the inverse demand curve is produced. The first demand curve is based on the initial values of pr , 2, and y , 1000. The second shows the result of increasing pr from 2 to 5, holding y constant. The third shows the effect of increasing y from 1000 to 1500, holding pr constant. The fourth shows the combined result of the two partially offsetting changes.

$$\begin{aligned} [pr0, y0] : [2, 1000] \$ \quad [pr1, y1] : [5, 1500] \$ \\ [a0, ED0, EDR0, EDY0] : [10, -0.8, -0.2, 0.4] \$ \\ [DP(q, a0, ED0, pr0, EDR0, y0, EDY0), \\ DP(q, a0, ED0, pr1, EDR0, y0, EDY0), \\ DP(q, a0, ED0, pr0, EDR0, y1, EDY0), \\ DP(q, a0, ED0, pr1, EDR0, y1, EDY0)] ; \\ \left[\frac{472.87}{q^{1.25}}, \frac{376.06}{q^{1.25}}, \frac{579.15}{q^{1.25}}, \frac{460.58}{q^{1.25}} \right] \end{aligned}$$

2.3.2.1 Income Elasticity

Most goods are normal goods, for which the demand changes in the same direction as income. These goods exhibit positive income elasticities of demand. Inferior goods have a negative income elasticity of demand.

Within normal goods, some goods have an income elasticity of demand that exceeds 1.0. These are sometimes called *luxury goods*. For a luxury good, spending on the good changes by a greater percentage than income, so the *share* of income spent on these goods varies directly with income.

For other goods, those for which the income elasticity of demand is less than 1.0 (inferior goods included), the share of income spent on the good changes inversely with respect to income level changes.

The example below illustrates a case in which the income elasticity of demand is less than one: $EDY0 = 0.4$. The income level increases by 50%, so the rightward shift in the demand curve should be about 20%. The value 162.27 in the second

a general way of specifying the response of a variable to a change, rather than a list of standard formulas.

demand curve equation below is 17.61 % larger than the value 137.97 in the first equation (recall that each of these is the number of units demanded when $p = 1$).

$$\left[\begin{array}{l} DQ(p, a0, ED0, pr0, EDR0, y0, EDY0) , \\ DQ(p, a0, ED0, pr0, EDR0, y1, EDY0) \end{array} \right] ; \quad \left[\frac{137.97}{p^{0.8}}, \frac{162.27}{p^{0.8}} \right]$$

Finally, we confirm that the increased income causes spending on this good to increase but that its share of total spending decreases. When $p = 3$ (any price could be used), buyers' spending equals 359.3 per period, 35.9 % of their income. When income increases to 1,500, spending increases to 440.06, only 29.3 % of the higher income level.¹⁶


```
EXP0:3*(DP(3,a0,ED0,pr0,EDR0,y0,EDY0))$
EXP1:3*(DP(3,a0,ED0,pr0,EDR0,y1,EDY0))$
matrix(["Income", "Spending", "Share"],
       [y0, EXP0, EXP0/y0], [y1, EXP1, EXP1/y1]) ;
```

Income Spending Share		
1000	359.3	0.359
1500	440.06	0.293

2.3.2.2 Cross-Price Elasticity

The cross-price elasticity measures the effect of a change in the price of a related good. The cross-price elasticity $EDR0 = -0.2$ implies that the two goods are complements: an increased price of the related good leads to a decreased demand for the good in question. As shown below, the higher price of the related good shifts the demand curve to the left (i.e., downward).

```
wxdraw2d(yrange=[0,500],xlabel="Quantity",ylabel=
"$ per unit", key = "Initial Demand",
explicit(DP(q,a0,ED0,pr0,EDR0,y0,EDY0),q,.1,5),
line_width=2, key="Demand, higher pr",
explicit(DP(q,a0,ED0,pr1,EDR0,y0,EDY0),q,.1,5))$
```



¹⁶For some specifications, the income shares might exceed 1.0. This result represents economic nonsense. If such a result occurs, reduce $a0$ and rerun the entire sheet, using `ctrl-r`.

2.3.2.3 Supply Elasticity

The concept of the price elasticity of supply differs little from that of its counterpart for demand. We consider it briefly in this section. The next section shows an important implication of these two elasticity measures taken as a pair, the incidence of excise taxes.


The input/output group below defines the equation, determines the equation of the supply curve for a set of parameters, and determines the quantity supplied given those parameters and a specific price.

$[SQ(p, b, ES) := b \cdot p^{ES}, \quad SP(q, b, ES) := (q/b)^{\frac{1}{ES}}];$
$[SQ(p, 40, 0.5), \quad SQ(1.5, 40, 0.5)];$

$$[SQ(p, b, ES) := b p^{ES}, \quad SP(q, b, ES) := \left(\frac{q}{b}\right)^{\frac{1}{ES}}] \quad [40 p^{0.5}, 48.99]$$

The graph below shows three supply curves. The curve exhibiting unitary elasticity is a ray through the origin.

```
wxdraw2d(user_preamble="set key top left", xlabel=
"Quantity", yrange=[0,2], ylabel="$ per unit", key=
"Unitary Elasticity", explicit(SP(q, 40, 1), q, 0, 75),
key="Inelastic", line_width=2, explicit(SP(q, 40, .5),
q, 0, 75), key="Elastic", color=gray,
explicit(SP(q, 40, 1.5), q, 0, 75) )$
```



When the constant-elasticity supply curve exhibits unitary elasticity, its graph is a straight line. We can show that the following is also true: A linear supply curve that intersects the price (vertical) axis is elastic at all points and that a linear supply curve that intersects the quantity (horizontal) axis is inelastic at all points. This is true regardless of the linear supply curve's slope.

Let $q = a + b \cdot p$. Then $\frac{dq}{dp} = b$, and the supply elasticity equals $b \cdot \frac{p}{q}$. The cell below shows that, therefore, the elasticity equals $\frac{b \cdot p}{b \cdot p + a}$. If $a > 0$, the supply curve intersects the price axis at $p > 0$, and the elasticity is less than 1.0. If $a < 0$, then the supply curve intersects the horizontal axis at a positive value, the number of units that would be supplied even if $p = 0$, and the elasticity exceeds 1.0.

The third output entry shows that as price and quantity increase (rightward movement along the linear supply curve), the price elasticity approaches 1.0, irrespective of a 's sign.

$$\left[q : a + b \cdot p, \quad b \cdot p / q, \quad \lim_{p \rightarrow \infty} (b \cdot p / q, p, \infty) \right];$$

$$\left[b \cdot p + a, \quad \frac{b \cdot p}{b \cdot p + a}, \quad 1 \right]$$

2.4 Taxes and Efficiency

2.4.1 Tax Incidence

An important implication of the demand/supply model is that a tax imposed on the sale or purchase of the product will be born in part by sellers and in part by buyers. Furthermore, the effects of the tax on buyers and sellers are independent of whether the tax is nominally placed on selling or on buying.

The expressions below are for the demand and supply curves. That is, all other variables are held constant, and their effects are embedded in the coefficients a and b . We consider an excise tax here, leaving the examination of an ad valorem tax as an exercise.

First, establish the equilibrium in the absence of taxes.

```
DQ(p,a,ED):=a*p^ED$ DP(q,a,ED):=(q/a)^(1/ED)$
SQ(p,b,ES):=b*p^ES$ SP(q,b,ES):=(q/b)^(1/ES)$
[a0, ED0,b0,ES0]:[1200,-4,40,0.75]$ qE:find_root(
  DP(q,a0,ED0)-SP(q,b0,ES0)=0,q,0.01,500)$
pE:find_root(DQ(p,a0,ED0)-SQ(p,b0,ES0)=0,
  p,0.01,500)$ [qE, pE];
```

[68.437, 2.0463]

The results of an excise tax, $X0 = 0.50$ are as follows: the quantity traded falls from 68.437 units to 57.82 units, the price paid by buyers rises from 2.0463 to 2.1344, and the price received by sellers falls from 2.0463 to 1.6344. That is, the price paid rises by 0.0881, and the price received falls by 0.412. Thus, for this particular demand and supply pair, the tax falls disproportionately on sellers.

The equality in expression $qX0$ shows the willingness-to-pay function on the left-hand side and the supply price on the right-hand side. The latter contains two parts: the original function, which reflects non-tax costs, and the per-unit excise tax.


```
X0:0.5$ qX0:find_root(DP(q,a0,ED0)=SP(q,b0,ES0)+X0,
  q,0.01,500)$ pDX0:DP(qX0,a0,ED0)$ pSX0:SP(
  qX0,b0,ES0)$ [qX0,pDX0,pSX0]; [pDX0-pE,pSX0-pE];
```

[57.82, 2.1344, 1.6344] [0.0881, -0.412]

If the tax had been placed on the buyer of the good rather than the seller of the good, the price that buyers would be willing and able to pay to sellers at each quantity would be the original willingness to pay *less* the tax, X_0 . Thus the left-hand side of the expression above would become $DP(q, a_0, ED_0) - X_0$ and the right-hand side would revert to $SP(q, b_0, ES_0)$. All solution values would be the same.

The graph shows the upward shift in the supply curve and the resulting equilibrium values. The supply curve shifts upward, parallel to the initial curve, by the amount of the excise tax, X_0 . If the tax had been placed on buyers, the demand curve would shift downward by X_0 , reflecting the lower net price that buyers are willing and able to pay to sellers.


```
pGrossNew : if q < qX0 then pDX0 else 0$
pNetNew: if q < qX0 then pSX0 else 0$
wxdraw2d(user_preamble="set key bottom right",
  line_type=dots,xlabel="Quantity",ylabel=
  "$ per unit",yrange=[0, 1.25*pDX0], explicit(
  pGrossNew,q,0,1.25*qE),explicit(pNetNew,q,
  0,1.25*qE),key="Demand",
  line_type=solid,explicit(DP(q,a0,ED0),
  q,0,1.25*qE),color="gray20",key=
  "Initial Supply",explicit(SP(q,b0,ES0),q,0,
  1.25*qE),line_width=2, key="Tax-laden Supply",
  explicit(X0+SP(q,b0,ES0),q,0,1.25*qE))$
```



2.4.2 Consumer Surplus

Both buyers and sellers gain from exchange. The demand/supply model illustrates this gain. These gains are given the names Consumer Surplus and Producer Surplus. We begin our development of Consumer Surplus by relating the demand curve to marginal value.

```
DP(q,ED) := q^(1/ED); wxdraw2d(
  xlabel="Quantity", xrange=
  [0,10],ylabel="$ per unit",
  explicit(DP(q,-1.5),q,0,1.5))$
```

$$DP(q, ED) := q^{\frac{1}{ED}}$$


The `wxdraw2d` command shows a graph of the function over the range 0–1.5. (Remember: The units can be any size, perhaps in millions per year.) The graph's vertical axis is truncated so that the marginal value of, say, unit 0.0000001 does not appear (it is \$46,416 in our normalized dollars).¹⁷

At each quantity, the height of the demand curve is the amount that some consumer is willing and able to pay for that unit. If we could sum all of these values, we would have a measure of value for a specified number of units of this good. Some practical difficulties appear, however.

- The marginal willingness to pay of the first few units might not be meaningful. Consider, for example, water. A single liter per week would not be enough to keep even one consumer alive, so any statement about how much consumers would pay for the first liter makes no sense. Fortunately, we seldom seek the total value of a good. Rather, we typically analyze a change that results in an increase or decrease in the quantity without reference to the first few units.
- A related problem is that some goods may not be available in continuously divisible units. We follow convention and assume goods are continuously divisible.¹⁸

¹⁷For this function, we can determine the area under the demand curve. We do not, because in general such an exercise lacks meaning. For some goods like water and food, the marginal value of the first few units is beyond measure.

¹⁸This convention is more reasonable than it might appear. Individuals who consume commodities like automobiles make two decisions, how many units to own and how often to buy new models. If a higher price of new good rises slightly and many individuals adjust by keeping their units a bit longer, then the market quantity demanded will decrease slightly, perhaps a few hundred units in a market in which millions of units are sold.

- We intentionally refer to the marginal willingness to pay as *a* measure of value, not *the* measure of value. This function can change if the income distribution changes, so using it as *the* measure would imply that the current income distribution is optimal.
- Taking willingness to pay as an appropriate measure of value requires that we accept a libertarian view of valuation: the consumer is judged to be the best judge of value. If paternalism is involved, the value that one might assign to a given quantity might be either higher or lower than the sum of all marginal willingness to pay.
- The buyers must receive all marginal benefits from consuming the product. That is, third parties, who do not purchase the good, must not gain or lose any benefit from the buyer's having purchased the good.¹⁹ Similarly, the producers must pay all the marginal costs from producing the product.
- The units must be distributed to those consumers who place the highest marginal value on them. This will routinely happen with markets. When other distribution methods are used, the value will be smaller than the sum of the values of the first n units.
- This measure of value is only approximately the same as measures derived from consumer theory. In some important cases, the approximation may be quite close.

The caveats listed above constitute warning against mechanistic use of this measure of value. They do not imply that it cannot be useful.

2.4.2.1 Value of Additional Units

Consider an example in which the quantity is to be changed from one unit to some other number of units (0.8 units in the illustration). The value lost due to decreasing the quantity from 1 to 0.8 is the area under the demand curve between 1.0 and 0.8. The graph below shows three possible demand curves, all passing through the point (1, 1). The reason for having three curves is that one often does not know the demand curve. One does, however, know the current quantity of the good. We also suppose that the analyst knows the marginal value when $q = 1$. Further, we suppose that the analyst has an estimate of demand elasticity. We call that estimate $ED0$. We arbitrarily consider two other values of the elasticity, $ED1 = 0.75 \cdot ED0$ and $ED2 = 1.5 \cdot ED0$.²⁰


¹⁹This caveat refers to a direct effect via the third party's utility function. Suppose that buyer A's purchase of the good precludes buyer B's purchase because A is willing and able to pay more than B. Then buyer B is in a sense worse off than if A had not been in the market, but this fact implies nothing about the validity of this measure of value.

²⁰Choose $ED0$ such that no elasticity equals -1.0 . The three elasticity values are part of the *Maxima* output.


```

ED0:-0.75$ [ED1,ED2]:[0.75*ED0,1.5*ED0]$ [q1,p1]:
[0.8,DP(q1,ED0)]$ phigh:max(DP(q1,ED1),
DP(q1,ED2))$ pmax: max(1, phigh)$
wxdraw2d(yrange=[0,1.2*pmax],xlabel="Quantity",
ylabel="$ per unit",user_preamble=
"set key bottom left",points_joined=true,
line_type=dots,points([[1,0],[1,1]]),
points([[q1,0],[q1,phigh]]),key="Median E",
line_type=solid,
explicit(DP(q,ED0),q,0,1.5*max(1,q1)),
color=gray40,key="Less Elastic",line_width=2,
explicit(DP(q,ED1),q,0,1.5*max(1,q1)),
key="More Elastic",line_width=3,
explicit(DP(q,ED2),q,0,1.5*max(1,q1)))$

```



The graph above suggests an attractive way to bracket the actual value of the units in question: develop a menu of parameters (in this illustration, the price elasticity) and determine the implication of each set of parameters on the menu. To achieve this in our illustration, we integrate each of the three functions, beginning at $q = 1$ and ending at $q = 0.8$. Integration is accomplished with the `integrate` command. Inspection suggests that the area under the curve over this range (representing a 20–25 % change in quantity, depending on the direction of the move) is not very sensitive to the elasticity value. The “More Elastic” curve is twice as elastic as the “Less Elastic” curve. Even so, the difference in the two measures is 10.56 %.

```

[[change1,change2,change3]:float([integrate(
DP(q,ED0),q,1,q1),integrate(DP(q,ED1),q,1,q1),
integrate(DP(q,ED2),q,1,q1)]), change2/change3];

```


```

[[-0.232, -0.244, -0.22], 1.1056]

```

To illustrate the concept of Consumer Surplus, consider one of the three demand curves above and suppose that the price is $p = 1$. The graph below shows the area that corresponds to Consumer Surplus. The value of units 0 through 1, measured as willingness to pay, is the area under the demand curve over that range (this value may not be definable for reasons indicated above). The rectangle is the amount of that value that buyers pay for the product, and the remaining area is Consumer Surplus.²¹

```
wxdraw2d(yrange=[0, 5*pmax],xlabel="Quantity",ylabel=
"$ per unit",user_preamble="set key bottom left",
line_type=dots,points_joined=true,color=gray10,
points([[0,1],[1,1]]),points([[1,0],[1,1]]),
label(["Consumer Surplus", 0.25, 2.5]),label(
["Amount paid", 0.2, 0.5]),line_type=solid,
explicit(DP(q,ED0), q, 0, 1.5*max(1,q1)) )$
```



2.4.2.2 Prices and Consumer Surplus

In a market setting, the question that faces the analyst is often this: What is the change in Consumer Surplus due to a change in the good's price? The material below illustrates how this question can be addressed. We use the range of prices that is defined above: $p = 1$ and $p = p1$. At $p = 1$, $q = 1$ by construction. For $p = p1$, the demand elasticity determines the quantity demanded.

```
DQ(p,ED) := p^ED;
[q10,q11,q12]:
[DQ(p1,ED0),DQ(p1,ED1),
DQ(p1,ED2)];
```


$DQ(p,ED) := p^{ED}$ [0.8, 0.846, 0.716]

²¹Note the use of labels for these areas.

Each label contains a list with three elements: the text that is to appear in the graph, the horizontal distance at which the label is to be centered, and the label's height. The alignment can be fine-tuned. See `label_alignment` in the *Maxima* user's guide.

The graph below is much like the one above. The points that are graphed have been changed, however, so that the trapezoidal area is defined by the two lines, a segment of the price axis, and a segment of the demand curve. As textbooks show, a price increase reduces Consumer Surplus in two ways: It increases the price that buyers must pay for the units that they still purchase at the higher price, and it reduces the quantity consumed so that buyers lose the Consumer Surplus on the marginal units. In the case illustrated here, the former effect is far larger than the latter one.

```
wxdraw2d( yrange = [0,1.2*pmax],xlabel="Quantity",
  ylabel="$ per unit",user_preamble=
  "set key bottom left",line_type=dots,points_joined=
  true,points([[0,1],[1,1]]),points([[0,p1],
  [max(q10,q11,q12),p1]]),line_type=solid,key=
  "Median E",explicit(DP(q,ED0),q,0,1.5*max(1,q10)),
  color=dark_grey,key="Less Elastic",line_width=2,
  explicit(DP(q,ED1),q,0,1.5*max(1,q11)),key=
  "More Elastic",color=gray40,
  explicit(DP(q,ED2),q,0,1.5*max(1,q12)))$
```



To determine the value of the change in Consumer Surplus for each of the three demand curves, we integrate the demand curve (not the inverse demand curve as above) over the range of prices. The three resulting values appear below. Again, the difference between the high-elasticity and the low-elasticity cases is small.²²

```
[change4, change5, change6]:float([integrate(
  DQ(p,ED0),p,1,p1),integrate(DQ(p, ED1),p,1,p1),
  integrate(DQ(p,ED2),p,1, p1)]); change5/change6;
```

[0.309,0.318,0.292]1.088

²²This result is not general. Suppose the demand curves were shifted horizontally to the left (and truncated where they enter the second quadrant). Then the change in Consumer Surplus would be smaller than appears above, and the percentage differences among the three cases would be larger.

2.4.3 Producer Surplus

Producer Surplus is analogous to Consumer Surplus. It is the difference between the revenue received by sellers and the revenue required to induce them to produce a given quantity (which is also the opportunity cost of producing that quantity of the good). We consider the case of an industry of price-taking firms. For such firms the height of the supply curve defines the minimum price required to bring a specified quantity to market.


For purposes of illustration, we define this function in a fashion similar to that used for demand. The only difference is that price and quantity units have been defined so that $(1, 1)$ is a point on the demand curve. For the same good, this implies that the supply curve must have a coefficient b so that it need not pass through $(1, 1)$. We denote the price elasticity of supply ES .

$$[SQ(p, ES, b) := b \cdot p^{ES}, \quad SP(q, ES, b) := (1/b) \cdot q^{1/ES}] ;$$

$$[SQ(p, ES, b) := b p^{ES}, SP(q, ES, b) := \frac{1}{b} q^{\frac{1}{ES}}]$$


The graph below shows Producer Surplus. The supply curve passes through the quantity $q = 1$ at a price determined by the income elasticity ($ES = 1$) and the coefficient ($b = 1.5$). These values are arbitrary. We use the resulting value, $p = 0.667$ repeatedly in drawing the graph, so we create a temporary binding of the name `ptemp` to this value. We use labels as above to indicate the areas that correspond to Producer Surplus and to the cost of producing the good in question. The height of the supply curve equals the marginal cost of the good, so Producer Cost is the area under the supply curve between $q = 0$ and $q = 1$. The graph does not show a change, but it is easy to see that a price change results in a changed Producer Surplus for the same reason that it results in a changed Consumer Surplus: producers receive a different payment per unit, and they sell a different number of units.

```
ptemp:SP(1,0.5,1.5)$ wxdraw2d(yrange=[0,1.5*ptemp],
xlabel="Quantity", ylabel="$ per unit",
user_preamble="set key top left",
label(["Producer Surplus",.4,.8*ptemp]),
label(["Producer Cost", 0.8, 0.2*ptemp]),
label_alignment=left,label(
["Revenue = Producer Surplus+Producer Cost",
0.05,1.4*ptemp]),points_joined=true,points(
[[0,ptemp],[1,ptemp]]),points([[1,0],[1,ptemp]]),
explicit(SP(q, 0.5,1.5),q,0,1.5*max(1,q1)))$
```



The next graph contains both demand and supply. It shows that the equilibrium quantity is just above 1.0 and that the equilibrium height of either the demand curve or the supply at the equilibrium quantity is just under 1.0.

```
wxdraw2d(yrange=[0,1.25*pmax],xlabel="Quantity",
  user_preamble="set key center right", ylabel=
"$ per unit",color=gray10,key="Demand",
explicit(SP(q,0.5,1.5),q,0,2), line_width=2,
key="Supply",explicit(DP(q, ED0), q, 0, 2) )$
```



We confirm below what the graph shows: the equilibrium quantity is 1.1293 units and the equilibrium price is \$0.85.

```
[q_e: find_root(SP(q,0.5,1.5)-DP(q,ED0),q,0.001,200),
[pd_e, ps_e] : [DP(q_e, ED0), SP(q_e, 0.5, 1.5)]];

[1.1293, [0.85, 0.85]]
```

2.4.4 Combined Surpluses and Efficiency

This section confirms that the sum of Consumer Surplus and Producer Surplus is maximized at the market equilibrium depicted above (where the demand and

supply curves intersect). Denote the sum of the two surpluses as `surpluses`. To determine the functional form of the variable, we integrate the relevant functions over values of q .

<pre>assume (ES > 0, ED < -1) \$ surpluses: integrate(DP(q, ED) - p + (p - SP(q, ES, b)), q);</pre>	$\frac{q^{\frac{1}{ED}+1}}{\frac{1}{ED}+1} - \frac{q^{\frac{1}{ES}+1}}{b(\frac{1}{ES}+1)}$
--------------------------------------------------------------------------------------------------------------	--------------------------------------------------------------------------------------------

The derivative of `surpluses` with respect to q appears below. The first term is the height of the demand curve and the second term is the height of the supply curve.²³

<pre>diff(surpluses, q);</pre>	$q^{\frac{1}{ED}} - \frac{q^{\frac{1}{ES}}}{b}$
--------------------------------	-------------------------------------------------

The result above says that the demand price *less* the supply price must equal zero, or that the two must be equal. Thus, the condition is that the quantity must be the one at which the demand and supply curves intersect.

The sufficient condition for having determined a maximum value is that the second derivative be negative. The expression below shows the second derivative of `surpluses`. Both numerators are positive, as are b and ES . Because the own-price elasticity of demand $ED < 0$, this expression must be negative.

<pre>diff2 : diff(surpluses, q, 2);</pre>	$\frac{q^{\frac{1}{ED}-1}}{ED} - \frac{q^{\frac{1}{ES}-1}}{bES}$
-------------------------------------------	------------------------------------------------------------------

Finally we apply the values used above to confirm the results for a particular set of parameters. The embedded list in the output shows the equality of the demand price and the supply price. The other term shows the value of the second derivative, confirming that it is negative. The quantity that yields maximum surplus has been found.

<pre>[[q_e^(1/ED0), q_e^(1/0.5)/1.5], subst([ED=ED0, ES=0.5, b=1.5, q=q_e], diff2)];</pre>	$[[0.85, 0.85], -2.5097]$
---------------------------------------------------------------------------------------------------	---------------------------

²³This result can be predicted by examining the expression to be integrated, `surpluses`. Because p is subtracted from the height of the demand function and added in the definition of Producer Surplus, the function is $DP(q, ED) - SP(q, ES, b)$. This function is integrated (the “antiderivative” is extracted) and then the derivate of the integral is extracted, returning the original function.

Further Readings

The table at in the preface lists chapters in microeconomics textbooks that relate to this chapter. The following can also be of value:

Baldani J, Bradfield J, Turner RW (1996) Mathematical economics. Dryden, Fort Worth, Chaps. 1–2.

Wainwright K, Chiang AC (2004) Fundamental methods of mathematical economics, 4th edn. McGraw-Hill, New York, Chaps. 1–2.

Microeconomic Theory and Computation
Applying the Maxima Open-Source Computer Algebra
System

Hammock, M.R.; Mixon, Jr., J.W.

2013, XIX, 385 p. 275 illus., 8 illus. in color., Hardcover

ISBN: 978-1-4614-9416-4